
NAG Library Routine Document

F08TPF (ZHPGVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

F08TPF (ZHPGVX) computes selected eigenvalues and, optionally, eigenvectors of a complex
generalized Hermitian-definite eigenproblem, of the form

Az ¼ �Bz; ABz ¼ �z or BAz ¼ �z;

where A and B are Hermitian, stored in packed format, and B is also positive definite. Eigenvalues and
eigenvectors can be selected by specifying either a range of values or a range of indices for the desired
eigenvalues.

2 Specification

SUBROUTINE F08TPF (ITYPE, JOBZ, RANGE, UPLO, N, AP, BP, VL, VU, IL, IU,
ABSTOL, M, W, Z, LDZ, WORK, RWORK, IWORK, JFAIL,
INFO)

&
&

INTEGER ITYPE, N, IL, IU, M, LDZ, IWORK(5*N), JFAIL(*),
INFO

&

REAL (KIND=nag_wp) VL, VU, ABSTOL, W(N), RWORK(7*N)
COMPLEX (KIND=nag_wp) AP(*), BP(*), Z(LDZ,*), WORK(2*N)
CHARACTER(1) JOBZ, RANGE, UPLO

The routine may be called by its LAPACK name zhpgvx.

3 Description

F08TPF (ZHPGVX) first performs a Cholesky factorization of the matrix B as B ¼ UHU , when
UPLO ¼ U or B ¼ LLH, when UPLO ¼ L . The generalized problem is then reduced to a standard
symmetric eigenvalue problem

Cx ¼ �x;

which is solved for the desired eigenvalues and eigenvectors; the eigenvectors are then backtransformed
to give the eigenvectors of the original problem.

For the problem Az ¼ �Bz, the eigenvectors are normalized so that the matrix of eigenvectors, Z,
satisfies

ZHAZ ¼ � and ZHBZ ¼ I;

where � is the diagonal matrix whose diagonal elements are the eigenvalues. For the problem
ABz ¼ �z we correspondingly have

Z�1AZ�H ¼ � and ZHBZ ¼ I;

and for BAz ¼ �z we have

ZHAZ ¼ � and ZHB�1Z ¼ I:

5 Arguments

1: ITYPE – INTEGER Input

On entry: specifies the problem type to be solved.

ITYPE ¼ 1
Az ¼ �Bz.

ITYPE ¼ 2
ABz ¼ �z.

ITYPE ¼ 3
BAz ¼ �z.

Constraint: ITYPE ¼ 1, 2 or 3.

2: JOBZ – CHARACTER(1) Input

On entry: indicates whether eigenvectors are computed.

JOBZ ¼ N
Only eigenvalues are computed.

JOBZ ¼ V
Eigenvalues and eigenvectors are computed.

Constraint: JOBZ ¼ N or V .

3: RANGE – CHARACTER(1) Input

On entry: if RANGE ¼ A , all eigenvalues will be found.

If RANGE ¼ V , all eigenvalues in the half-open interval VL;VUð � will be found.

If RANGE ¼ I , the ILth to IUth eigenvalues will be found.

Constraint: RANGE ¼ A , V or I .

4: UPLO – CHARACTER(1) Input

On entry: if UPLO ¼ U , the upper triangles of A and B are stored.

If UPLO ¼ L , the lower triangles of A and B are stored.

Constraint: UPLO ¼ U or L .

5: N – INTEGER Input

On entry: n, the order of the matrices A and B.

Constraint: N � 0.

6: APð�Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the dimension of the array AP must be at least max 1;N� Nþ 1ð Þ=2ð Þ.
On entry: the upper or lower triangle of the n by n Hermitian matrix A, packed by columns.

More precisely,

if UPLO ¼ U , the upper triangle of A must be stored with element Aij in
APðiþ j j� 1ð Þ=2Þ for i � j;

if UPLO ¼ L , the lower triangle of A must be stored with element Aij in
APðiþ 2n� jð Þ j� 1ð Þ=2Þ for i � j.

On exit: the contents of AP are destroyed.

7: BPð�Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the dimension of the array BP must be at least max 1;N� Nþ 1ð Þ=2ð Þ.
On entry: the upper or lower triangle of the n by n Hermitian matrix B, packed by columns.

More precisely,

if UPLO ¼ U , the upper triangle of B must be stored with element Bij in
BPðiþ j j� 1ð Þ=2Þ for i � j;

if UPLO ¼ L , the lower triangle of B must be stored with element Bij in
BPðiþ 2n� jð Þ j� 1ð Þ=2Þ for i � j.

On exit: the triangular factor U or L from the Cholesky factorization B ¼ UHU or B ¼ LLH, in
the same storage format as B.

8: VL – REAL (KIND=nag_wp) Input
9: VU – REAL (KIND=nag_wp) Input

On entry: if RANGE ¼ V , the lower and upper bounds of the interval to be searched for
eigenvalues.

If RANGE ¼ A or I , VL and VU are not referenced.

Constraint: if RANGE ¼ V , VL < VU.

10: IL – INTEGER Input
11: IU – INTEGER Input

On entry: if RANGE ¼ I , the indices (in ascending order) of the smallest and largest eigenvalues
to be returned.

If RANGE ¼ A or V , IL and IU are not referenced.

Constraints:

if RANGE ¼ I and N ¼ 0, IL ¼ 1 and IU ¼ 0;
if RANGE ¼ I and N > 0, 1 � IL � IU � N.

12: ABSTOL – REAL (KIND=nag_wp) Input

On entry: the absolute error tolerance for the eigenvalues. An approximate eigenvalue is accepted
as converged when it is determined to lie in an interval a; b½ � of width less than or equal to

ABSTOLþ �max aj j; bj jð Þ;
where � is the machine precision. If ABSTOL is less than or equal to zero, then � Tk k1 will be
used in its place, where T is the tridiagonal matrix obtained by reducing C to tridiagonal form.
Eigenvalues will be computed most accurately when ABSTOL is set to twice the underflow
threshold 2� X02AMFð Þ, not zero. If this routine returns with INFO ¼ 1 to N, indicating that
some eigenvectors did not converge, try setting ABSTOL to 2� X02AMFð Þ. See Demmel and
Kahan (1990).

13: M – INTEGER Output

On exit: the total number of eigenvalues found. 0 � M � N.

If RANGE ¼ A , M ¼ N.

If RANGE ¼ I , M ¼ IU� ILþ 1.

14: WðNÞ – REAL (KIND=nag_wp) array Output

On exit: the first M elements contain the selected eigenvalues in ascending order.

15: ZðLDZ; �Þ – COMPLEX (KIND=nag_wp) array Output

Note: the second dimension of the array Z must be at least max 1;Mð Þ if JOBZ ¼ V , and at least
1 otherwise.

On exit: if JOBZ ¼ V , then

if INFO ¼ 0, the first M columns of Z contain the orthonormal eigenvectors of the matrix
A corresponding to the selected eigenvalues, with the ith column of Z holding the
eigenvector associated with WðiÞ. The eigenvectors are normalized as follows:

if ITYPE ¼ 1 or 2, ZHBZ ¼ I;

if ITYPE ¼ 3, ZHB�1Z ¼ I;

if an eigenvector fails to converge (INFO ¼ 1 to N), then that column of Z contains the
latest approximation to the eigenvector, and the index of the eigenvector is returned in
JFAIL.

If JOBZ ¼ N , Z is not referenced.

Note: you must ensure that at least max 1;Mð Þ columns are supplied in the array Z; if
RANGE ¼ V , the exact value of M is not known in advance and an upper bound of at least N
must be used.

16: LDZ – INTEGER Input

On entry: the first dimension of the array Z as declared in the (sub)program from which F08TPF
(ZHPGVX) is called.

Constraints:

if JOBZ ¼ V , LDZ � max 1;Nð Þ;
otherwise LDZ � 1.

17: WORKð2� NÞ – COMPLEX (KIND=nag_wp) array Workspace

18: RWORKð7� NÞ – REAL (KIND=nag_wp) array Workspace

19: IWORKð5� NÞ – INTEGER array Workspace

20: JFAILð�Þ – INTEGER array Output

Note: the dimension of the array JFAIL must be at least max 1;Nð Þ.
On exit: if JOBZ ¼ V , then

if INFO ¼ 0, the first M elements of JFAIL are zero;

if INFO ¼ 1 to N, JFAIL contains the indices of the eigenvectors that failed to converge.

If JOBZ ¼ N , JFAIL is not referenced.

21: INFO – INTEGER Output

On exit: INFO ¼ 0 unless the routine detects an error (see Section 6).

F08TPF NAG Library Manual

F08TPF.4 Mark 26



6 Error Indicators and Warnings

INFO < 0

If INFO ¼ �i, argument i had an illegal value. An explanatory message is output, and execution
of the program is terminated.

INFO ¼ 1 to N

If INFO ¼ i, F08GPF (ZHPEVX) failed to converge; i eigenvectors failed to converge. Their
indices are stored in array JFAIL.

INFO > N

F07GRF (ZPPTRF) returned an error code; i.e., if INFO ¼ Nþ i, for 1 � i � N, then the leading
minor of order i of B is not positive definite. The factorization of B could not be completed and
no eigenvalues or eigenvectors were computed.

7 Accuracy

If B is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and
vectors may be large, although when the diagonal elements of B differ widely in magnitude the
eigenvalues and eigenvectors may be less sensitive than the condition of B would suggest. See Section
4.10 of Anderson et al. (1999) for details of the error bounds.

8 Parallelism and Performance

F08TPF (ZHPGVX) is threaded by NAG for parallel execution in multithreaded implementations of the
NAG Library.

F08TPF (ZHPGVX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the
vendor library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is proportional to n3.

The real analogue of this routine is F08TBF (DSPGVX).

10 Example

This example finds the eigenvalues in the half-open interval �3; 3ð �, and corresponding eigenvectors, of
the generalized Hermitian eigenproblem Az ¼ �Bz, where

A ¼
�7:36 0:77� 0:43i �0:64� 0:92i 3:01� 6:97i
0:77þ 0:43i 3:49 2:19þ 4:45i 1:90þ 3:73i

�0:64þ 0:92i 2:19� 4:45i 0:12 2:88� 3:17i
3:01þ 6:97i 1:90� 3:73i 2:88þ 3:17i �2:54

0
B@

1
CA

and

B ¼
3:23 1:51� 1:92i 1:90þ 0:84i 0:42þ 2:50i
1:51þ 1:92i 3:58 �0:23þ 1:11i �1:18þ 1:37i
1:90� 0:84i �0:23� 1:11i 4:09 2:33� 0:14i
0:42� 2:50i �1:18� 1:37i 2:33þ 0:14i 4:29

0
B@

1
CA:

The example program for F08TQF (ZHPGVD) illustrates solving a generalized symmetric
eigenproblem of the form ABz ¼ �z.

10.1 Program Text

Program f08tpfe

! F08TPF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: nag_wp, x04daf, zhpgvx

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: zero = 0.0E+0_nag_wp
Integer, Parameter :: nin = 5, nout = 6
Character (1), Parameter :: uplo = ’U’

! .. Local Scalars ..
Complex (Kind=nag_wp) :: scal
Real (Kind=nag_wp) :: abstol, vl, vu
Integer :: i, ifail, il, info, iu, j, k, ldz, &

m, n
! .. Local Arrays ..

Complex (Kind=nag_wp), Allocatable :: ap(:), bp(:), work(:), z(:,:)
Real (Kind=nag_wp), Allocatable :: rwork(:), w(:)
Integer, Allocatable :: iwork(:), jfail(:)

! .. Intrinsic Procedures ..
Intrinsic :: abs, conjg, maxloc

! .. Executable Statements ..
Write (nout,*) ’F08TPF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)
Read (nin,*) n
ldz = n
m = n
Allocate (ap((n*(n+1))/2),bp((n*(n+1))/2),work(2*n),z(ldz,m),rwork(7*n), &

w(n),iwork(5*n),jfail(n))

! Read the lower and upper bounds of the interval to be searched,
! and read the upper or lower triangular parts of the matrices A
! and B from data file

Read (nin,*) vl, vu
If (uplo==’U’) Then

Read (nin,*)((ap(i+(j*(j-1))/2),j=i,n),i=1,n)
Read (nin,*)((bp(i+(j*(j-1))/2),j=i,n),i=1,n)

Else If (uplo==’L’) Then
Read (nin,*)((ap(i+((2*n-j)*(j-1))/2),j=1,i),i=1,n)
Read (nin,*)((bp(i+((2*n-j)*(j-1))/2),j=1,i),i=1,n)

End If

! Set the absolute error tolerance for eigenvalues. With abstol
! set to zero, the default value is used instead

abstol = zero

! Solve the generalized Hermitian eigenvalue problem
! A*x = lambda*B*x (itype = 1)

! The NAG name equivalent of zhpgvx is f08tpf
Call zhpgvx(1,’Vectors’,’Values in range’,uplo,n,ap,bp,vl,vu,il,iu, &

abstol,m,w,z,ldz,work,rwork,iwork,jfail,info)

If (info>=0 .And. info<=n) Then

! Print solution

Write (nout,99999) ’Number of eigenvalues found =’, m
Write (nout,*)
Write (nout,*) ’Eigenvalues’
Write (nout,99998) w(1:m)
Flush (nout)

! Normalize the eigenvectors, largest element real
Do i = 1, m

rwork(1:n) = abs(z(1:n,i))
k = maxloc(rwork(1:n),1)
scal = conjg(z(k,i))/abs(z(k,i))
z(1:n,i) = z(1:n,i)*scal

End Do

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call x04daf(’General’,’ ’,n,m,z,ldz,’Selected eigenvectors’,ifail)

If (info>0) Then
Write (nout,99999) ’INFO eigenvectors failed to converge, INFO =’, &

info
Write (nout,*) ’Indices of eigenvectors that did not converge’
Write (nout,99997) jfail(1:m)

End If
Else If (info>n .And. info<=2*n) Then

i = info - n
Write (nout,99996) ’The leading minor of order ’, i, &

’ of B is not positive definite’
Else

Write (nout,99999) ’Failure in ZHPGVX. INFO =’, info
End If

99999 Format (1X,A,I5)
99998 Format (3X,(8F8.4))
99997 Format (3X,(8I8))
99996 Format (1X,A,I4,A)

End Program f08tpfe

10.2 Program Data

F08TPF Example Program Data

4 :Value of N

-3.0 3.0 :Values of VL and VU

(-7.36, 0.00) ( 0.77, -0.43) (-0.64, -0.92) ( 3.01, -6.97)
( 3.49, 0.00) ( 2.19, 4.45) ( 1.90, 3.73)

( 0.12, 0.00) ( 2.88, -3.17)
(-2.54, 0.00) :End of matrix A

( 3.23, 0.00) ( 1.51, -1.92) ( 1.90, 0.84) ( 0.42, 2.50)
( 3.58, 0.00) (-0.23, 1.11) (-1.18, 1.37)

( 4.09, 0.00) ( 2.33, -0.14)
( 4.29, 0.00) :End of matrix B

10.3 Program Results

F08TPF Example Program Results

Number of eigenvalues found = 2

Eigenvalues
-2.9936 0.5047

Selected eigenvectors
1 2

1 -0.6626 0.2835
0.2258 -0.5806

2 -0.1164 -0.3769
-0.0178 -0.3194

3 0.9098 -0.3338
-0.0000 -0.0134

4 -0.6120 0.6663
-0.5348 0.0000

